
9. Beispiele für Anwendungsprogrammierung
In diesem Kapitel finden Sie Programmierbeispiele für verschiedene Anwendungsszenarien.

9.1 Ein Roboter pro Zuführung und ein Teil pro Zuführung

9.1.1 Programmbeispiel 1.1 

Beispieltyp: 
Verwendung von PF_Robot-Callback für die Bewegung

Konfiguration
Anzahl Roboter: 1
Anzahl Zuführungen: 1
Anzahl Teiletypen auf der Zuführung: 1
Anzahl Ablagepositionen: 1
Kameraausrichtung: fest nach unten ausgerichtet montierte Kamera über Zuführung 1

Beschreibung
Die Teile werden von der Zuführung aufgenommen und in eine Box gelegt. Sobald alle Teile aufgenommen wurden, fordert der Control-Callback weitere Teile an. Wenn weitere Teile durch den Bediener oder aus dem Bunker bereitgestellt werden, wird der Zyklus fortgesetzt. Durch Ausführung des PF_Stop-Befehls wird der aktuelle Zyklus beendet und die Anwendung abgeschlossen. 
(Der PF_Stop-Befehl wird im folgenden Beispielcode nicht verwendet.)

Beispielcode 
Main.prg

Function main
	If Not Motor = On Then
		Motor On
	EndIf
	Power Low
	Jump park
	PF_Start 1
Fend



PartFeeding.prg

Function PF_Robot(PartID As Integer) As Integer
	Do While PF_QueLen(PartID) > 0
		P0 = PF_QueGet(PartID)
		Jump P0
		On Gripper; Wait 0.2;
		Jump Place
		Off Gripper; Wait 0.2;
		PF_QueRemove PartID
		If PF_IsStopRequested(PartID) = True Then
			Exit Do
		EndIf
	Loop
	PF_Robot = PF_CALLBACK_SUCCESS
	
Fend


9.1.2 Programmbeispiel 1.2

Beispieltyp: 
[bookmark: _Hlk47009615]System mit mehreren Robotern – 1 Roboter pro Zuführung und 1 Teil pro Zuführung

Konfiguration
Anzahl Roboter: 2. Roboter 1 ist an die Steuerung RC700A angeschlossen und Roboter 2 verwendet eine Drive Unit
Anzahl Zuführungen: 2
Anzahl Teiletypen auf jeder Zuführung: 1
Anzahl Ablagepositionen: 1 pro Roboter
Kameraausrichtung: fest nach unten ausgerichtet montierte Kamera über jeder Zuführung 

Beschreibung
Beide Roboter nehmen unabhängig voneinander Teile auf und legen diese ab. Die Roboter arbeiten weder an derselben Zuführung noch legen sie Teile an derselben Ablageposition ab. Die Arbeitsbereiche der Roboter überschneiden sich nicht. Beide Roboter haben Punkte mit den Labels „Park“, „Pick“ und „Place“ in den jeweiligen Punktedateien gespeichert.

Beispielcode
Main.prg

Function main
	Robot 1
	If Motor = Off Then Motor On
	Power Low
	Jump Park

	Robot 2
	If Motor = Off Then Motor On
	Power Low
	Jump Park

	PF_Start 1
PF_Start 2
Fend



PartFeeding.prg

Function PF_Robot(PartID As Integer) As Integer
	Integer gripperOutput

	Select PartID
		Case 1
			Robot 1
			gripperOutput =1
Case 2
			Robot 2
			gripperOutput =2
	Send
	
Do While PF_QueLen(PartID) > 0
		Pick = PF_QueGet(PartID)
		Jump Pick
		On gripperOutput; Wait 0.2;
		Jump Place
		Off gripperOutput; Wait 0.2;
		PF_QueRemove PartID
		If PF_IsStopRequested(PartID) = True Then
			Exit Do
		EndIf
	Loop
	PF_Robot = PF_CALLBACK_SUCCESS
	
Fend


9.1.3 Programmbeispiel 1.3 

Beispieltyp:
Bildverarbeitung und Vibration parallel zur Roboterbewegung

Konfiguration
Anzahl Roboter: 1
Anzahl Zuführungen: 1
Anzahl Teiletypen: 1 
Anzahl Ablagepositionen: 1
Kameraausrichtung: fest nach unten ausgerichtet montierte Kamera

Beschreibung
Der Roboter nimmt ein Teil mit der Teilenummer 1 von der Zuführung auf und setzt es in eine Halterung ein. Dieser Vorgang wird wiederholt, bis alle nach oben gerichteten Teile von der Zuführung aufgenommen wurden. Während das letzte verfügbare Teil abgelegt wird (90 % der Bewegung abgeschlossen), vibriert die Zuführung, der Bunker stellt ggf. weitere Teile bereit usw.
Dieses Beispiel zeigt, wie der Betrieb der Zuführung parallel zur Roboterbewegung erfolgen kann, um so den Durchsatz zu optimieren. 

Beispielcode
Main.prg

Function main
	MemOff PartsToPick
	Off Gripper
	
	Robot 1
	If Motor = Off Then Motor On
	Power Low
	Jump Park
	
	PF_Start(1)				
		
	Xqt RobotPickPlace	
Fend


Function RobotPickPlace
	DO
Wait MemSw(PartsToPick) = On
If PF_QueLen(1) > 0 Then
		Do 
				Pick = PF_QueGet(1) 
				PF_QueRemove 1
				Jump Pick
				On Gripper; Wait 0.2
				If PF_QueLen(1) = 0 Then
					Jump Place ! D90; MemOff PartsToPick!
					Off Gripper; Wait 0.2
					Exit Do
				Else
					Jump Place
					Off Gripper; Wait 0.2
				EndIf
			Loop
		Else
			MemOff PartsToPick	
		EndIf
Loop
Fend

PartFeeding.prg

Function PF_Robot(PartID As Integer) As Integer
	MemOn PartsToPick
	Wait MemSw(PartsToPick) = Off

	PF_Robot = PF_CALLBACK_SUCCESS
Fend	


9.1.4 Programmbeispiel 1.4 

Beispieltyp: 
1 Roboter, 2 Zuführungen und 1 Teil pro Zuführung

Konfiguration
Anzahl Roboter: 1
Anzahl Zuführungen: 2
Anzahl Teiletypen auf jeder Zuführung: 1
Anzahl Ablagepositionen: 1
Kameraausrichtung: fest nach unten ausgerichtet montierte Kamera über jeder Zuführung

Beschreibung
Der Roboter nimmt ein Teil mit der Teilenummer 1 von Zuführung 1 auf und legt es in eine Box. Dieser Vorgang wird wiederholt, bis alle nach oben gerichteten Teile von Zuführung 1 aufgenommen wurden. Anschließend nimmt der Roboter alle nach oben gerichteten Teile von Zuführung 2 auf und legt sie in die Box. Die Kameras und Zuführungen arbeiten jeweils parallel.

Beispielcode
Main.prg

Function main
	MemOff PartsToPick1; MemOff PartsToPick2
	Off Gripper
	
	Robot 1
	If Motor = Off Then Motor On
	Power Low
	Jump Park
	
	PF_Start(1)				
PF_Start(2)	
		
	Xqt rbt1	
Fend


Function rbt1	
	Do
		Call RobotPickPlace(1)			
Call RobotPickPlace(2)
Loop
Fend

Function RobotPickPlace(PartID As Integer)
	Integer partsToPickMembit,	
	
Select PartID
		Case 1
			partsToPickMembit = IONumber("PartsToPick1")		
Case 2
			partsToPickMembit = IONumber("PartsToPick2")	
	Send
	
	Wait MemSw(partsToPickMembit) = On
If PF_QueLen(PartID) > 0 Then

		Do 
			Pick = PF_QueGet(PartID) 
			PF_QueRemove PartID
			Jump Pick
			On Gripper; Wait 0.2
			If PF_QueLen(PartID) = 0 Then
				Jump Place ! D90; MemOff partsToPickMembit !
				Off Gripper; Wait 0.2
				Exit Do
			Else
				Jump Place
				Off Gripper; Wait 0.2
			EndIf
		
			If PF_IsStopRequested(PartID) = True Then
				MemOff partsToPickMembit	
Exit Do
			EndIf
		Loop
	Else
		MemOff partsToPickMembit	
	EndIf
Fend

PartFeeding.prg
Function PF_Robot(PartID As Integer) As Integer
	
	Select PartID
		Case 1
			MemOn PartsToPick1
			Wait MemSw(PartsToPick1) = Off
		Case 2
			MemOn PartsToPick2
			Wait MemSw(PartsToPick2) = Off
	Send

	PF_Robot = PF_CALLBACK_SUCCESS
Fend	


9.1.5 Programmbeispiel 1.5 

Beispieltyp: 
1 Roboter, 2 Zuführungen und 1 Teil pro Zuführung – mobile Kamera

Konfiguration
Anzahl Roboter: 1
Anzahl Zuführungen: 2
Anzahl Teiletypen auf jeder Zuführung: 1
Anzahl Ablagepositionen: 1
Kameraausrichtung:	Beide Zuführungen nutzen die am Roboter montierte mobile Kamera
(in diesem Beispiel werden keine fest nach unten ausgerichtet montierten Kameras verwendet)

Beschreibung
Der Roboter bewegt die mobile Kamera über Zuführung 1 und nimmt ein Bild auf. Der Roboter nimmt alle Teile von Zuführung 1 auf und legt sie in eine Box. Dieser Vorgang wird wiederholt, bis alle nach oben gerichteten Teile von Zuführung 1 aufgenommen wurden. Anschließend bewegt der Roboter die mobile Kamera über Zuführung 2 und nimmt ein Bild auf.
Der Roboter nimmt alle korrekt ausgerichteten Teile von Zuführung 2 auf und legt sie in die Box.

Beispielcode
Main.prg

Function main
	MemOff PartsToPick1; MemOff PartsToPick2;
	MemOff mobileCamBefore1; MemOff mobileCamAfter1
	MemOff mobileCamBefore2; MemOff mobileCamAfter2
	MemOff mobileCamInPos1; MemOff mobileCamInPos2
	
	Robot 1
	If Motor = Off Then Motor On
	Power Low
	Jump Park
	
	PF_Start(1)			
	PF_Start(2)			
	Xqt rbt1			
Fend



Function rbt1
	Do
		Wait MemSw(mobileCamBefore1) = On
		Jump MobileCamShotFeeder1; MemOn mobileCamInPos1
		Wait MemSw(mobileCamAfter1) = On
		MemOff mobileCamInPos1

		Call RobotPickPlace(1)	
		
		Wait MemSw(mobileCamBefore2) = On
		Jump MobileCamShotFeeder2; MemOn mobileCamInPos2
		Wait MemSw(mobileCamAfter2) = On
		MemOff mobileCamInPos2

		Call RobotPickPlace(2)	
	Loop
Fend

Function RobotPickPlace(PartID As Integer)
	Integer partsToPickMembit, partCnt, gripperOutput, toolNum
	
Select PartID
		Case 1
			partsToPickMembit = IONumber("PartsToPick1")		
Case 2
			partsToPickMembit = IONumber("PartsToPick2")	
	Send
	
	Wait MemSw(partsToPickMembit) = On 
	Do While PF_QueLen(PartID) > 0
		P0 = PF_QueGet(PartID)
		Jump P0
		On Gripper; Wait 0.2;
		Jump Place
		Off Gripper; Wait 0.2;
		PF_QueRemove PartID
		If PF_IsStopRequested(PartID) = True Then
			Exit Do
		EndIf
	Loop
	MemOff partsToPickMembit	
Fend


PartFeeding.prg

Function PF_Robot(PartID As Integer) As Integer
	
	Select PartID
		Case 1
			MemOn PartsToPick1
			Wait MemSw(PartsToPick1) = Off
		Case 2
			MemOn PartsToPick2
			Wait MemSw(PartsToPick2) = Off
	Send

	PF_Robot = PF_CALLBACK_SUCCESS
Fend

Function PF_MobileCam(PartID As Integer, Action As Integer) As Integer
Integer mobileCamBeforeMembit, mobileCamAfterMembit, mobileCamInPosMembit
	
	Select PartID
		Case 1
			mobileCamBeforeMembit = IONumber("mobileCamBefore1")							mobileCamAfterMembit = IONumber("mobileCamAfter1")							mobileCamInPosMembit = IONumber("mobileCamInPos1")
		Case 2
			mobileCamBeforeMembit = IONumber("mobileCamBefore2")							mobileCamAfterMembit = IONumber("mobileCamAfter2")							mobileCamInPosMembit = IONumber("mobileCamInPos2")
	Send
	
	Select Action
		Case PF_MOBILECAM_BEFORE 
' Anforderung an den Roboter, in die Kameraposition zu fahren
			MemOff mobileCamAfterMembit
			MemOn mobileCamBeforeMembit
			Wait MemSw(mobileCamInPosMembit) = On
		Case PF_MOBILECAM_AFTER 
' Anforderung an den Roboter, sich nach der Bilderfassung des Teils zu bewegen
			MemOff mobileCamBeforeMembit
			MemOn mobileCamAfterMembit
			Wait MemSw(mobileCamInPosMembit) = Off
	Send

	PF_MobileCam = PF_CALLBACK_SUCCESS
Fend


9.1.6 Programmbeispiel 1.6 

Beispieltyp: 
Festlegen der Anzahl aufzunehmender Teile

Konfiguration
Anzahl Roboter: 1
Anzahl Zuführungen: 2
Anzahl Teiletypen auf jeder Zuführung: 1
Anzahl Ablagepositionen: 1
Kameraausrichtung: fest nach unten ausgerichtet montierte Kamera über jeder Zuführung 

Beschreibung
Der Roboter nimmt vier Teile mit der Teilenummer 1 von Zuführung 1 auf und legt sie einzeln ab. Anschließend nimmt der Roboter nimmt fünf Teile mit der Teilenummer 2 von Zuführung 2 auf und legt sie ab. Die Kameras und Zuführungen arbeiten jeweils parallel. Dieses Beispiel veranschaulicht, wie der Roboter eine bestimmte Anzahl von Teilen von jeder Zuführung aufnehmen kann. Der Beispielcode ermöglicht auch die Auswahl aller verfügbaren Teile, indem der Parameter numToPick auf ALL_AVAILABLE gesetzt wird. Wenn nach dem Entnehmen der gewünschten Menge noch nach oben gerichtete Teile auf einer der beiden Zuführungen liegen, vibriert die Zuführung nicht unnötig.

Beispielcode
Main.prg

#define ALL_AVAILABLE -1

Function main
	MemOff PartsToPick1; MemOff PartsToPick2
	Off Gripper
	
	Robot 1
	If Motor = Off Then Motor On
	Power Low
	Jump Park

	PF_Start(1)			
	PF_Start(2)			
	
Xqt rbt1			
Fend


Function rbt1
Do
		Call RobotPickPlace(1, 4)	'part 1...pick & place 4 times
		Call RobotPickPlace(2, 5)	'part 2...pick & place 5 times
	Loop
Fend

Function RobotPickPlace(PartID As Integer, numToPick As Integer)
	Integer partsToPickMembit, partCnt, 
	
Select PartID
		Case 1
			partsToPickMembit = IONumber("PartsToPick1")		
Case 2
			partsToPickMembit = IONumber("PartsToPick2")	
	Send
	
	partCnt = 0
	Do
		Wait MemSw(partsToPickMembit) = On
		If PF_QueLen(PartID) > 0 Then
			Pick = PF_QueGet(PartID) 
			PF_QueRemove PartID
			Jump Pick
			On Gripper; Wait 0.2
			partCnt = partCnt + 1
			If PF_QueLen(PartID) = 0 Then
				Jump Place ! D90; MemOff partsToPickMembit !
				Off Gripper; Wait 0.2
				If (partCnt = numToPick) Or (numToPick = ALL_AVAILABLE) Then
					Exit Do
				EndIf
			Else
				Jump Place
				Off Gripper; Wait 0.2
				If (partCnt = numToPick) Then
					Exit Do
				EndIf
			EndIf
		Else
			MemOff partsToPickMembit
		EndIf
		If PF_IsStopRequested(PartID) = True Then
			MemOff partsToPickMembit
			Exit Do
		EndIf
	Loop
Fend


PartFeeding.prg

Function PF_Robot(PartID As Integer) As Integer
	
	Select PartID
		Case 1
			MemOn PartsToPick1
			Wait MemSw(PartsToPick1) = Off
		Case 2
			MemOn PartsToPick2
			Wait MemSw(PartsToPick2) = Off
	Send

	PF_Robot = PF_CALLBACK_SUCCESS
Fend	


9.1.7 Programmbeispiel 1.7 

Beispieltyp: 
Sicherstellen, dass der Roboter der für das Teil verwendeten Roboternummer entspricht

Konfiguration
Anzahl Zuführungen: mehr als 1 
Anzahl Teiletypen auf jeder Zuführung: 1
Kameraausrichtung: fest nach unten ausgerichtet montierte Kamera über jeder Zuführung

Beschreibung
Wenn Sie mehrere Zuführungen verwenden, ist es üblich, dass der Roboter die Bewegung in einem Multi-Task und nicht innerhalb des PF_Robot-Callbacks ausführt. Mit anderen Worten: Der PF_Robot-Callback lädt einfach Teilepositionen (Punkte) in den Teilezuführungs-Queue. Ein Roboter-Bewegungs-Task nutzt die Punkte im Queue. Wenn der Code auf diese Weise aufgebaut ist, sollte der Roboter-Bewegungs-Task prüfen, ob die aktuelle Roboternummer mit der Roboternummer übereinstimmt, die für das Teil ausgewählt wurde. Durch diese zusätzliche Überprüfung wird sichergestellt, dass der Roboter nicht die falschen Punkte verwendet und es zu keiner Kollision kommt.

[bookmark: _Hlk47025726]Beispielcode
Main.prg

Function RobotPickPlace(PartID As Integer, numToPick As Integer)

	If PF_Info(PartID, PF_INFO_ID_ROBOT_NO) <> Robot Then
Print "Robot does not match the robot # for the current part"
		Quit All
	EndIf

	‘Roboter-Bewegungscode
Fend


9.1.8 Programmbeispiel 1.8 

Beispieltyp: 
Neues Bild erfassen und Queue nach jedem Pick-Vorgang laden

Konfiguration
Anzahl Roboter: 1
Anzahl Zuführungen: 1
Anzahl Teiletypen: 1 
Anzahl Ablagepositionen: 1
Kameraausrichtung: fest nach unten ausgerichtet montierte Kamera

Beschreibung
Der Roboter nimmt ein Teil mit der Teilenummer 1 von der Zuführung auf und setzt es in eine Halterung ein. Nach jedem Pick-and-Place-Vorgang wird ein neues Bild erfasst und der Queue neu geladen. In diesem Beispiel soll verhindert werden, dass umliegende Teile beim Aufnehmen gestört werden. Der PF_Robot-Callback-Rückgabewert PF_CALLBACK_RESTART erzwingt eine erneute Bildverarbeitung für alle Teile und ein erneutes Laden aller Teile-Queues. Dieses Vorgehen ist zwar nicht effizient, aber durch die Erfassung eines Bildes in jedem Zyklus kann PF_CALLBACK_RESTART gegebenenfalls dazu beitragen, die Genauigkeit zu verbessern. Die Bilderfassung und die Vibration der Zuführung erfolgen parallel zur Roboterbewegung. 

[bookmark: _Hlk47025771]Beispielcode
Main.prg

Function main
	MemOff PartsToPick
	Off Gripper
	
	Robot 1
	If Motor = Off Then Motor On
	Power Low
	Jump Park
	
	PF_Start(1)				
		
	Xqt RobotPickPlace	
Fend


Function RobotPickPlace
	Do	
Wait MemSw(PartsToPick) = On
If PF_QueLen(1) > 0 Then 
			Pick = PF_QueGet(1) 
			PF_QueRemove 1
			Jump Pick
			On Gripper; Wait 0.2
			Jump Place ! D90; MemOff PartsToPick!
			Off Gripper; Wait 0.2
		Else
			MemOff PartsToPick	
		EndIf
	Loop
Fend

PartFeeding.prg

Function PF_Robot(PartID As Integer) As Integer
	MemOn PartsToPick
	Wait MemSw(PartsToPick) = Off

	PF_Robot = PF_CALLBACK_RESTART
Fend	


9.1.9 Programmbeispiel 1.9 

Beispieltyp: 
Teil nach Ausrichtung (Vorder-/Rückseite) sortieren

Konfiguration
Anzahl Roboter: 1
Anzahl Zuführungen: 1
Anzahl Teiletypen auf der Zuführung: 1
Anzahl Ablagepositionen: 2 (eine für die Vorderseite und eine für die Rückseite)
Kameraausrichtung: fest nach unten ausgerichtet montierte Kamera über Zuführung 1

Seite [Allgemeines] für Teil 1:
[image: ]

Seite [Vision] für Teil 1:
[image: ]

Ablageposition für nach oben gerichtete Teile

Beschreibung
Bei dieser Anwendung sortiert der Roboter die Teile entsprechend der Ausrichtung von Vorder- und Rückseite. Nach oben gerichtete Teile werden an einem Punkt mit dem Label „PlaceFront“ und nach unten gerichtete Teile an einem anderen Punkt mit dem Label „PlaceBack“ abgelegt. Die Reihenfolge, in der die Teile aufgenommen werden, ist bei dieser Anwendung irrelevant. Der Roboter nimmt und platziert so viele nach oben gerichtete Teile, wie er kann, und nimmt und platziert anschließend so viele nach unten gerichtete Teile, wie er kann. Wenn „Wenden erforderlich“ deaktiviert ist, aber Vision-Objekte unter „Bildobjekt für die Vorderseite des Teils“ und „Bildobjekt für die Rückseite des Teils“ ausgewählt wurden, lädt das System sowohl nach oben als auch nach unten gerichtete Teile in den Koordinaten-Queue und setzt den Wert für die Teileausrichtung entsprechend. Bitte beachten Sie, dass dem System durch die Einstellung „Wenden erforderlich“ mitgeteilt wird, dass die Teile in einer bestimmten Ausrichtung aufgenommen werden sollen. Durch Deaktivieren dieser Einstellung teilen Sie dem System mit, dass Teile mit beiden Ausrichtungen aufgenommen werden sollen.

In diesem Fall wird die Ausrichtung der nach oben gerichteten Teile im Teilekoordinaten-Queue automatisch auf die Konstante PF_PARTORIENT_FRONT gesetzt. Die Ausrichtung der nach unten gerichteten Teile wird automatisch auf die Konstante PF_PARTORIENT_BACK gesetzt. 


[bookmark: _Hlk47025801]Beispielcode 
Main.prg

Function main
	If Not Motor = On Then
		Motor On
	EndIf
	Power Low
	Jump park
	PF_Start 1
Fend

PartFeeding.prg

Function PF_Robot(PartID As Integer) As Integer
	Do While PF_QueLen(PartID) > 0
		P0 = PF_QueGet(PartID)
		Jump P0
		On Gripper; Wait 0.2;
If PF_QuePartOrient(PartID) = PF_PARTORIENT_FRONT Then
			Jump PlaceFront
		ElseIf PF_QuePartOrient(PartID) = PF_PARTORIENT_BACK Then
			Jump PlaceBack
		EndIf
		Off Gripper; Wait 0.2;
		PF_QueRemove PartID
		If PF_IsStopRequested(PartID) = True Then
			Exit Do
		EndIf
	Loop
	PF_Robot = PF_CALLBACK_SUCCESS
	
Fend



9.1.10 Programmbeispiel 1.10 

Beispieltyp: 
Verwendung des PF_Vision-Callback 

Konfiguration
Anzahl Roboter: 1
Anzahl Zuführungen: 1
Anzahl Teiletypen auf der Zuführung: 1
Anzahl Ablagepositionen: 1
Kameraausrichtung: fest nach unten ausgerichtet montierte Kamera über Zuführung 1

Beschreibung
[bookmark: _Hlk47009896]Dieses Beispiel zeigt, wie sich der PF_Vision-Callback verwenden lässt, um ein Bild zu erfassen und den Teilekoordinaten-Queue mit den Bildverarbeitungsergebnissen zu laden. Um diese Funktion zu verwenden, wählen Sie im EPSON RC+ 7.0-Menü unter [Tools]-[Teilezuführung]-[Vision] die Option „Der Benutzer verarbeitet das Bild für das Teil per PF_Vision-Callback“.
[bookmark: _Hlk47025836]
Beispielcode 
Main.prg

Function main
	If Not Motor = On Then
		Motor On
	EndIf
	Power Low
	Jump park
	PF_Start 1
Fend



PartFeeding.prg

Function PF_Robot(PartID As Integer) As Integer
	Do While PF_QueLen(PartID) > 0
		P0 = PF_QueGet(PartID)
		Jump P0
		On Gripper; Wait 0.2;
		Jump Place
		Off Gripper; Wait 0.2;
		PF_QueRemove PartID
		If PF_IsStopRequested(PartID) = True Then
			Exit Do
		EndIf
	Loop
	PF_Robot = PF_CALLBACK_SUCCESS
	
Fend

Function PF_Vision(PartID As Integer, ByRef numBack As Integer) As Integer
	Boolean found
	Integer i, numFront
	Real RB_X, RB_Y, RB_U, RB_Z
	
	' Pick Z coordinate
	RB_Z = -132.0
	
	' Initialize coordinates queue
	PF_QueRemove PartID, All
	
	 PF_Backlight 1, On
	 
	' Detect the parts 
	VRun UsrVisionSeq
	
	PF_Backlight 1, Off
 
	VGet UsrVisionSeq.Geom01.NumberFound, numFront	‘Nach oben gerichtete Teile
	VGet UsrVisionSeq.Geom02.NumberFound, numBack	‘Nach unten gerichtete Teile
	If numFront <> 0 Then
		For i = 1 To numFront
	 	  	VGet UsrVisionSeq.Geom01.RobotXYU(i), found, RB_X, RB_Y, RB_U
			If found Then
				PF_QueAdd PartID, XY(RB_X, RB_Y, RB_Z, RB_U)
			EndIf
		Next
	EndIf
	
	PF_Vision = PF_CALLBACK_SUCCESS
	
Fend


9.1.11 Programmbeispiel 1.11 

Beispieltyp: 
Aufnehmen eines Teils mit unterschiedlichen Seiten

Konfiguration
Anzahl Roboter: 1
Anzahl Zuführungen: 1
Anzahl Teiletypen auf der Zuführung: 1
Anzahl unterschiedlicher Seiten des Teils: 3
Anzahl Ablagepositionen: 3 (eine Ablageposition für jede Seite des Teils)
Kameraausrichtung: fest nach unten ausgerichtet montierte Kamera über Zuführung 1

Ablageposition für Ausrichtung „rechts“

Beschreibung
In diesem Beispiel liegt ein physischer Teiletyp auf der Zuführung. Das Teil hat drei unterschiedliche Seiten (rechts, links und oben). Der Roboter kann das Teil in jeder der drei Ausrichtungen aufnehmen. Aus der Sicht des Vision-Systems ist jede Seite ein anderes Teil (obwohl es sich eigentlich um ein physisches Teil handelt). Die Option Teilezuführung ermöglicht die gleichzeitige Verarbeitung von vier unterschiedlichen Teilen auf derselben Zuführung. Wir legen daher für jede Seite ein einzelnes Teil an. Dazu erstellen wir drei Vision-Sequenzen für das Teil und benennen sie „links“, „rechts“ und „oben“. Jede Vision-Sequenz verwendet ein Geometric-Objekt, um das Teil in seiner spezifischen Ausrichtung zu lokalisieren. Im Dialog [Teilezuführung] erstellen wir drei Teile – Teil 1, 2 und 3. Teil 1 lokalisiert das Teil mit der Ausrichtung „links“. Teil 2 lokalisiert das Teil mit der Ausrichtung „rechts“. Teil 3 lokalisiert das Teil mit der Ausrichtung „oben“. Die Option „Wenden erforderlich“ wird bei allen drei Teilen deaktiviert. Alle drei Teile verwenden dieselbe Blob-Bildsequenz des Teils mit der Bezeichnung „PartBlob“. 

[bookmark: _Hlk47013575][image: ]

Lernen Sie den Z-Wert für das Aufnehmen jedes einzelnen Teils ein (die Aufnahmehöhe kann je nach Seite des Teils unterschiedlich sein). Teile mit der Ausrichtung „links“ werden an einem Punkt mit dem Label „PlaceLeft“ abgelegt. Teile mit der Ausrichtung „rechts“ werden an einem Punkt mit dem Label „PlaceRight“ abgelegt. Teile mit der Ausrichtung „oben“ werden an einem Punkt mit dem Label „PlaceTop“ abgelegt. Der Roboter nimmt und platziert zunächst alle Teile mit der Ausrichtung „links“, anschließend alle Teile mit der Ausrichtung „rechts“ und zuletzt alle Teile mit der Ausrichtung „oben“. 
[bookmark: _Hlk47025879]
Beispielcode 
Main.prg

Function main
	If Not Motor = On Then Motor On
	Power Low
	Jump park
	PF_Start 1,2,3
Fend

PartFeeding.prg

Function PF_Robot(PartID As Integer) As Integer
	
	Do While PF_QueLen(PartID) > 0
		P10 = PF_QueGet(PartID)
		Jump P10
		On Vacuum; Wait 0.2
		Select PartID
			Case 1
				Jump PlaceLeft
			Case 2
				Jump PlaceRight
			Case 3
				Jump PlaceTop
		Send
		Off Vacuum; Wait 0.2
		PF_QueRemove PartID
	Loop
	
	PartID = PartID + 1
	If PartID > 3 Then PartID = 1
	PF_ActivePart PartID
	PF_Robot = PF_CALLBACK_SUCCESS
	
Fend


Dieses Beispiel zeigt eine einfache Methode, um Teile mit unterschiedlichen Seiten durch das Anlegen mehrerer Teile zu verarbeiten. Ebenso ist es auch möglich, die Option „Der Benutzer verarbeitet das Bild“ zu wählen und den PF_Vision-Callback zu nutzen. Durch Verwendung von PF_QueAdd können Benutzerdaten in den Teilekoordinaten-Queue aufgenommen werden. Diese Benutzerdaten können beispielsweise numerische Werte sein, die die Teileausrichtung angeben (z. B. 1 = links, 2 = rechts, 3 = oben). In diesem Fall wird die Funktion PF_QueUserData dazu verwendet, den Ausrichtungswert zu ermitteln, damit das Teil an der richtigen Stelle abgelegt werden kann. Im Code müssten dann auch die Höhenunterschiede der unterschiedlichen Seiten des Teils berücksichtigt werden.
HINWEIS



9.2 Ein Roboter – mehrere Teile

9.2.1 Programmbeispiel 2.1 

Beispieltyp: 
1 Roboter und mehrere Teile – Bildverarbeitung und Vibration parallel zur Roboterbewegung

Konfiguration
Anzahl Roboter: 1
Anzahl Zuführungen: 1
Anzahl Teiletypen auf der Zuführung: 2
Anzahl Ablagepositionen: 1
Kameraausrichtung: fest nach unten ausgerichtet montierte Kamera 
[image: ]Roboter 1

Beschreibung
[bookmark: _Hlk47013800]Es liegen zwei unterschiedliche Teiletypen auf der Zuführung. Der Roboter nimmt kontinuierlich zwei Teile mit der Nummer 1 auf und legt sie ab und nimmt anschließend ein Teil mit der Nummer 2 auf und legt es ab. Dies wird durch den ständigen Wechsel von PF_ActivePart erreicht. Bei dieser Anwendung ist die Reihenfolge, in der die Teile aufgenommen werden, entscheidend (beispielweise bei einem Montagevorgang). Sobald das letzte Teil abgelegt wurde, wird PF_ActivePart geändert, um das gewünschte Teil zuzuführen und dem System zu signalisieren, dass es gegebenenfalls vibrieren und Bilder erfassen soll. Dieser Vorgang erfolgt parallel zur Roboterbewegung (wenn 30 % der Wegstrecke zur Ablageposition zurückgelegt wurden). 
[bookmark: _Hlk47025914]
Beispielcode
Main.prg

Function Main
	Integer numToPick1, numToPick2, i
	
	Robot 1
	Motor On
	Power High
	Speed 50
	Accel 50, 50
	Jump park

	MemOff PartsToPick1
	MemOff PartsToPick2
	numToPick1 = 2
	numToPick2 = 1
	
	PF_Start 1,2
	

	Do
		i = 0
		Do
			Wait MemSw(PartsToPick1) = On
			P0 = PF_QueGet(1)
			PF_QueRemove (1)
			Jump P0 /R
			On Gripper
			Wait .25
			i = i + 1
			If i < numToPick1 And PF_QueLen(1) > 0 Then
				Jump place
			Else
				'Letztes Teil bzw. keine weiteren Teile, die aufgenommen werden können
				If i = numToPick1 Then
					PF_ActivePart 2
             EndIf
				Jump place ! D30; MemOff PartsToPick1 !
			EndIf
			Off Gripper
			Wait .25
        Loop Until i = numToPick1
		i = 0
		Do
			Wait MemSw(PartsToPick2) = On
			P0 = PF_QueGet(2)
			PF_QueRemove (2)
			Jump P0 /R
			On Gripper
			Wait .25
			i = i + 1
			If i < numToPick2 And PF_QueLen(2) > 0 Then
				Jump place
			Else
				'Letztes Teil bzw. keine weiteren Teile, die aufgenommen werden können
				If i = numToPick2 Then
					PF_ActivePart 1
				EndIf
				Jump place ! D30; MemOff PartsToPick2 !
			EndIf
			Off Gripper
			Wait .25
		Loop Until i = numToPick2
	Loop 
Fend

PartFeeding.prg

Function PF_Robot(PartID As Integer) As Integer
	Select PartID
		Case 1
			MemOn PartsToPick1
			Wait MemSw(PartsToPick1) = Off
		Case 2
			MemOn PartsToPick2
			Wait MemSw(PartsToPick2) = Off
	Send

	PF_Robot = PF_CALLBACK_SUCCESS
Fend


9.3 Zwei Roboter – ein Teil

9.3.1 Programmbeispiel 3.1 

Beispieltyp: 
2 Roboter und 1 physischer Teiletyp – Bewegung im PF_Robot-Callback – Aufnehmen in einer bestimmten Reihenfolge

Konfiguration
Anzahl Roboter: 2
Anzahl Zuführungen: 1
Anzahl Teiletypen auf der Zuführung: 1
Anzahl Ablagepositionen: 2
Kameraausrichtung: fest nach unten ausgerichtet montierte Kamera 
[image: ]Ablageposition für Roboter 2

Beschreibung
Das System besteht aus zwei Robotern und einer Zuführung. Es gibt nur einen physischen Teiletypen. Da jeder Roboter seine eigene Kamerakalibrierung hat, gibt es zwei logische Teile – Teil 1 für Roboter 1 und Teil 2 für Roboter 2. Die Roboter nehmen abwechselnd Teile von der Zuführung auf. Die Teile müssen bei dieser Anwendung in einer bestimmten Reihenfolge aufgenommen werden. Die abwechselnde Verarbeitung der beiden Teile wird mithilfe von PF_ActivePart erzielt. Die Roboterbewegung erfolgt innerhalb des PF_Robot-Callback. In diesem Beispiel erfolgt die Roboterbewegung nicht parallel zum Betrieb der Zuführung. Der Code ist einfach, aber nicht effizient. Jeder Roboter hat einen Punkt mit dem Label „park“ und einen Punkt mit dem Label „place“.
Der entscheidende Wert in diesem Beispiel ist der PF_Robot-Callback-Rückgabewert PF_CALLBACK_RESTART_ACTIVEPART. Dieser Rückgabewert ermöglicht es mehreren Robotern, dieselbe Zuführung zu verwenden, ohne dass die Teilekoordinaten in den Queues beider Teile erscheinen. Der Rückgabewert erzwingt die Aufnahme eines neuen Bildes ausschließlich für das aktive Teil („PF_ActivePart“) und es wird nur der Queue des aktiven Teils geladen. 


[bookmark: _Hlk47025951]Beispielcode
Main.prg

Function Main
	Robot 1
	Motor On
	Power High
	Speed 50
	Accel 50, 50
	Jump park
	Robot 2
	Motor On
	Power High
	Speed 50
	Accel 50, 50
	Jump park

	PF_Start 1, 2
Fend

PartFeeding.prg

Function PF_Robot(PartID As Integer) As Integer
	If PF_QueLen(PartID) > 0 Then
		Select PartID
			Case 1
				Robot 1
				P0 = PF_QueGet(1)
				PF_QueRemove (1)
				Jump P0 /R
				On rbt1Gripper
				Wait .25
		       Jump place
				Off rbt1Gripper
				Wait .25
				PF_ActivePart 2
			Case 2
				Robot 2
				P0 = PF_QueGet(2)
				PF_QueRemove (2)
				Jump P0 /L
				On rbt2Gripper
				Wait .25
				Jump place
				Off rbt2Gripper
				Wait .25
				PF_ActivePart 1
		Send
	
	EndIf
		
	PF_Robot = PF_CALLBACK_RESTART_ACTIVEPART

Fend


9.3.2 Programmbeispiel 3.2 

Beispieltyp: 
[bookmark: _Hlk47013881]2 Roboter und 1 physischer Teiletyp – Bewegung in separaten Tasks – Reihenfolge irrelevant


Konfiguration
Anzahl Roboter: 2
Anzahl Zuführungen: 1
Anzahl Teiletypen auf der Zuführung: 1
Anzahl Ablagepositionen: 2
Kameraausrichtung: fest nach unten ausgerichtet montierte Kamera 

Beschreibung
Das System besteht aus zwei Robotern und einer Zuführung. Es gibt nur einen physischen Teiletypen. Da jeder Roboter seine eigene Kamerakalibrierung hat, gibt es zwei logische Teile – Teil 1 für Roboter 1 und Teil 2 für Roboter 2. Die Reihenfolge, in der die Teile aufgenommen werden, ist irrelevant und beruht auf dem Prinzip „wer zuerst kommt, wird zuerst bedient“. Das Besondere an diesem Beispiel ist, dass für jedes Teil in jedem Zyklus Bilder erfasst werden. Dies kann sinnvoll sein, um zu verhindern, dass umliegende Teile beim Aufnehmen gestört werden. Der PF_Robot-Callback-Rückgabewert PF_CALLBACK_RESTART erzwingt eine erneute Bildverarbeitung für alle Teile und ein erneutes Laden aller Teile-Queues.
Dieses Vorgehen ist zwar nicht effizient, aber PF_CALLBACK_RESTART kann unter bestimmten Umständen nützlich sein.


[bookmark: _Hlk47025978]Beispielcode
Main.prg

Function Main
	Robot 1
	Motor On
	Power High
	Speed 50
	Accel 50, 50
	Jump park
	Robot 2
	Motor On
	Power High
	Speed 50
	Accel 50, 50
	Jump park
	MemOff PartsToPick
	PF_Start 1,2
	Xqt Robot1PickPlace
	Xqt Robot2PickPlace
Fend

Function Robot1PickPlace
	Robot 1
	
	Do
		PF_AccessFeeder(1)
		Wait MemSw(PartsToPick) = On 
If PF_QueLen(1) > 0 Then
			P0 = PF_QueGet(1)
			PF_QueRemove (1)
			Jump P0 /R
			On 5
			Wait .5
          Jump place ! D30; MemOff PartsToPick; PF_ReleaseFeeder 1!
			Off 5
			Wait .25
		Else
			MemOff PartsToPick; PF_ReleaseFeeder 1
		EndIf
	Loop
Fend

Function Robot2PickPlace
	Robot 2
	
	Do
		PF_AccessFeeder(1)
		Wait MemSw(PartsToPick) = On 
If PF_QueLen(2) > 0 Then
			P0 = PF_QueGet(2)
			PF_QueRemove (2)
			Jump P0 /L
			On 2
			Wait .5
          Jump place ! D30; MemOff PartsToPick; PF_ReleaseFeeder 1!
			Off 2
			Wait .25
		Else
			MemOff PartsToPick; PF_ReleaseFeeder 1
		EndIf
	Loop
Fend

PartFeeding.prg

Function PF_Robot(PartID As Integer) As Integer
	MemOn PartsToPick
	Wait MemSw(PartsToPick) = Off
	
	
	PF_Robot = PF_CALLBACK_RESTART 'Erzwingt erneute Bildverarbeitung und Vibration
Fend


9.3.3 Programmbeispiel 3.3

Beispieltyp: 
2 Roboter und 1 physischer Teiletyp – Bewegung in separaten Tasks – „wer zuerst kommt, wird zuerst bedient“ – simulierte Prozessverzögerung

Konfiguration
Anzahl Roboter: 2
Anzahl Zuführungen: 1
Anzahl Teiletypen auf der Zuführung: 1
Anzahl Ablagepositionen: 2
Kameraausrichtung: fest nach unten ausgerichtet montierte Kamera 

Beschreibung
Das System besteht aus zwei Robotern und einer Zuführung. Es gibt nur einen physischen Teiletypen. Da jeder Roboter seine eigene Kamerakalibrierung hat, gibt es zwei logische Teile – Teil 1 für Roboter 1 und Teil 2 für Roboter 2. In diesem Beispiel hat jeder Roboter eine variable Verarbeitungszeit (simuliert durch eine beliebige Wartezeit). Jeder Roboter ist nach dem Aufnehmen eines Teils von der Zuführung damit beschäftigt, einen anderen Vorgang auszuführen. Mithilfe der Merkerbits „Rbt1Complete“ und „Rbt2Complete“ wird signalisiert, dass der Roboter seine Sekundäroperation beendet hat und nun bereit ist, ein anderes Teil von der Zuführung aufzunehmen. Der PF_Robot-Callback gibt den Wert PF_CALLBACK_RESTART_ACTIVEPART zurück, wenn das gewünschte Teil (PF_ActivePart) nicht mit dem aktuellen Teil übereinstimmt (d. h. wenn der andere Roboter ein Teil aufnehmen will). Dadurch wird verhindert, dass die Punkte mehrfach in den Roboter-Queues erscheinen. Es wird ein neues Bild für das aktive Teil („PF_ActivePart“) aufgenommen und der Queue des aktiven Teils wird neu geladen. Wenn jedoch das nächste Teil mit dem aktuellen Teil übereinstimmt (d. h. derselbe Roboter wird ein Teil von der Zuführung aufnehmen), dann lautet der Rückgabewert des PF_Robot-Callbacks PF_CALLBACK_SUCCESS. Mithilfe von PF_AccessFeeder und PF_ReleaseFeeder wird sichergestellt, dass die Roboter nicht kollidieren, wenn sie auf die Zuführung zugreifen.


[bookmark: _Hlk47026001]Beispielcode
Main.prg

Function Main
	Robot 1
	Motor On
	Power High
	Speed 50
	Accel 50, 50
	Jump place
	Robot 2
	Motor On
	Power High
	Speed 50
	Accel 50, 50
	Jump place
	MemOff PartsToPick1
	MemOff PartsToPick2
	
	PF_Start 1,2
	Xqt Robot1PickPlace
	Xqt Robot2PickPlace
Fend

Function Robot1PickPlace
	Integer randomTime
	
	Robot 1
	MemOn Rbt1Complete
	
	Do
		Wait MemSw(PartsToPick1) = On
		PF_AccessFeeder(1)
		MemOff Rbt1Complete
		P0 = PF_QueGet(1)
		PF_QueRemove (1)
		Jump P0 /R
		On rbt1Gripper
		Wait .25
     	Jump place ! D30; MemOff PartsToPick1; PF_ReleaseFeeder 1!
		Off rbt1Gripper
		Wait .25
		'Testet lange Verarbeitungszeit – Roboter ist beschäftigt 
		Randomize
		randomTime = Int(Rnd(9)) + 1
		Wait randomTime
		MemOn Rbt1Complete
	Loop
Fend


[bookmark: _Hlk47026020]Function Robot2PickPlace
	Integer randomTime
	
	Robot 2
	MemOn Rbt2Complete
	
	Do
		Wait MemSw(PartsToPick2) = On
		PF_AccessFeeder(1)
		MemOff Rbt2Complete
		P0 = PF_QueGet(2)
		PF_QueRemove (2)
		Jump P0 /L
		On rbt2Gripper
		Wait .25
		Jump place ! D30; MemOff PartsToPick2; PF_ReleaseFeeder 1!
		Off rbt2Gripper
		Wait .25
		'Testet lange Verarbeitungszeit – Roboter ist beschäftigt 
		Randomize
		randomTime = Int(Rnd(9)) + 1
		Wait randomTime
		MemOn Rbt2Complete
	Loop
Fend

PartFeeding.prg

Function PF_Robot(PartID As Integer) As Integer
Integer nextPart
 
	Select PartID
		Case 1
			MemOn PartsToPick1
          Wait MemSw(PartsToPick1) = Off
		Case 2
			MemOn PartsToPick2
			Wait MemSw(PartsToPick2) = Off
	Send

	Wait MemSw(Rbt1Complete) = On Or MemSw(Rbt2Complete) = On
	If MemSw(Rbt1Complete) = On Then
		nextPart = 1
	ElseIf MemSw(Rbt2Complete) = On Then
		nextPart = 2
	EndIf

	PF_ActivePart nextPart

	If nextPart = PartID Then
		'Identisches Teil, daher muss kein neues Bild aufgenommen und der Queue nicht neu geladen werden
		PF_Robot = PF_CALLBACK_SUCCESS
	Else
		'Neustart der Bildverarbeitung
'Erfasst Bild und lädt den Queue nur für das aktive Teil	
		PF_Robot = PF_CALLBACK_RESTART_ACTIVEPART
	EndIf
		
Fend
9.4 Zwei Roboter – mehrere Teile

9.4.1 Programmbeispiel 4.1

Beispieltyp: 
2 Roboter, 1 Zuführung, mehrere Teile – Bewegung im PF_Robot-Callback – Aufnehmen in einer bestimmten Reihenfolge

Konfiguration
Anzahl Roboter: 2
Anzahl Zuführungen: 1
Anzahl Teiletypen auf der Zuführung: 2
Anzahl Ablagepositionen: 2
Kameraausrichtung: fest nach unten ausgerichtet montierte Kamera 

Beschreibung
Das System besteht aus zwei Robotern und einer Zuführung. Jeder Roboter nimmt einen unterschiedlichen Teiletyp auf. Die Roboter nehmen abwechselnd Teile von der Zuführung auf. Die Teile müssen bei dieser Anwendung in einer bestimmten Reihenfolge aufgenommen werden. Die abwechselnde Verarbeitung der beiden Teile wird mithilfe von PF_ActivePart erzielt. Die Roboterbewegung erfolgt innerhalb des PF_Robot-Callback. In diesem Beispiel erfolgt die Roboterbewegung nicht parallel zum Betrieb der Zuführung. Der Code ist einfach, aber nicht effizient. Roboter 1 nimmt und platziert Teil 1. Roboter 2 nimmt und platziert Teil 2. Jeder Roboter hat einen Punkt mit dem Label „park“ und einen Punkt mit dem Label „place“. 



[bookmark: _Hlk47026035]Beispielcode
Main.prg

Function Main
	Robot 1
	Motor On
	Power High
	Speed 50
	Accel 50, 50
	Jump park
	Robot 2
	Motor On
	Power High
	Speed 50
	Accel 50, 50
	Jump park

	PF_Start 1, 2
Fend

PartFeeding.prg

Function PF_Robot(PartID As Integer) As Integer
	If PF_QueLen(PartID) > 0 Then
		Select PartID
			Case 1
				Robot 1
				P0 = PF_QueGet(1)
				PF_QueRemove (1)
				Jump P0 /R
				On rbt1Gripper
				Wait .25
		       Jump place
				Off rbt1Gripper
				Wait .25
				PF_ActivePart 2
			Case 2
				Robot 2
				P0 = PF_QueGet(2)
				PF_QueRemove (2)
				Jump P0 /L
				On rbt2Gripper
				Wait .25
				Jump place
				Off rbt2Gripper
				Wait .25
				PF_ActivePart 1
		Send
	
	EndIf
	
	PF_Robot = PF_CALLBACK_SUCCESS
	
Fend


9.4.2 Programmbeispiel 4.2

Beispieltyp: 
2 Roboter, 1 Zuführung, mehrere Teile – Bewegung in separaten Tasks – Aufnehmen in einer bestimmten Reihenfolge

Konfiguration
Anzahl Roboter: 2
Anzahl Zuführungen: 1
Anzahl Teiletypen auf der Zuführung: 2
Anzahl Ablagepositionen: 2
Kameraausrichtung: fest nach unten ausgerichtet montierte Kamera 

Beschreibung
Das System besteht aus zwei Robotern und einer Zuführung. Jeder Roboter nimmt einen unterschiedlichen Teiletyp auf. Die Roboter nehmen abwechselnd Teile von der Zuführung auf. Dies wird durch den ständigen Wechsel von PF_ActivePart erreicht. Wenn ein Roboter keine Teile zum Entnehmen mehr hat, darf der andere Roboter weiter Teile aus der Zuführung entnehmen, bis auch für diesen Roboter keine Teile mehr verfügbar sind. Roboter 1 nimmt und platziert Teil 1. Roboter 2 nimmt und platziert Teil 2. Jeder Roboter hat einen Punkt mit dem Label „park“ und einen Punkt mit dem Label „place“. Mithilfe von PF_AccessFeeder und PF_ReleaseFeeder wird verhindert, dass beide Roboter gleichzeitig versuchen, auf die Zuführung zuzugreifen. Sobald ein Roboter 30 % der Wegstrecke zu seiner Ablageposition abgeschlossen hat, darf der andere Roboter auf die Zuführung zugreifen.


[bookmark: _Hlk47026057]Beispielcode
Main.prg

Function Main
	Robot 1
	Motor On
	Power High
	Speed 50
	Accel 50, 50
	Jump park
	Robot 2
	Motor On
	Power Low
	Speed 50
	Accel 50, 50
	Jump park
	MemOff PartsToPick1
	MemOff PartsToPick2
	
	PF_Start 1, 2
	Xqt Robot1PickPlace
	Xqt Robot2PickPlace
Fend

Function Robot1PickPlace
	Robot 1
	Do
		Wait MemSw(PartsToPick1) = On
		PF_AccessFeeder 1
		P0 = PF_QueGet(1)
		PF_QueRemove (1)
		Jump P0 /R
		On 5
		Wait .5
		Jump place ! D30; MemOff PartsToPick1; PF_ReleaseFeeder 1 !
		Off 5
		Wait .25
	Loop
Fend

Function Robot2PickPlace
	Robot 2
	Do
		Wait MemSw(PartsToPick2) = On
		PF_AccessFeeder 1
		P0 = PF_QueGet(2)
		PF_QueRemove (2)
		Jump P0 /L
		On 2
		Wait .5
		Jump place ! D30; MemOff PartsToPick2; PF_ReleaseFeeder 1 !
		Off 2
		Wait .25
	Loop
Fend


PartFeeding.prg

Function PF_Robot(PartID As Integer) As Integer
	Select PartID
		Case 1
			If PF_QueLen(1) > 0 Then
				MemOn PartsToPick1
				Wait MemSw(PartsToPick1) = Off
				PF_ActivePart 2
			Else
				PF_ActivePart 1
			EndIf
		Case 2
			If PF_QueLen(2) > 0 Then
				MemOn PartsToPick2
				Wait MemSw(PartsToPick2) = Off
				PF_ActivePart 1
			Else
				PF_ActivePart 2
			EndIf
	Send
	PF_Robot = PF_CALLBACK_SUCCESS
Fend

9.4.3 Programmbeispiel 4.3

Beispieltyp: 
2 Roboter, 1 Zuführung, mehrere Teile – Bewegung in separaten Tasks – Aufnehmen in einer bestimmten Reihenfolge

Konfiguration
Anzahl Roboter: 2
Anzahl Zuführungen: 1
Anzahl Teiletypen auf der Zuführung: 3
Anzahl Ablagepositionen: 2
Kameraausrichtung: fest nach unten ausgerichtet montierte Kamera 

Beschreibung
Das System besteht aus zwei Robotern und einer Zuführung. Jeder Roboter nimmt ein unterschiedliches Teil auf. Roboter 1 nimmt und platziert ein Teil mit der Nummer 1. Anschließend nimmt und platziert Roboter 2 ein Teil mit der Nummer 4 und ein Teil mit der Nummer 5. Die Teile müssen bei dieser Anwendung in einer bestimmten Reihenfolge aufgenommen werden. Die abwechselnde Verarbeitung der beiden Teile wird mithilfe von PF_ActivePart erzielt. Die Roboterbewegung erfolgt parallel zur Vibration der Zuführung. 

[bookmark: _Hlk47026120]
Beispielcode
Main.prg

Function Main
	Robot 1
	Motor On
	Power High
	Speed 50
	Accel 50, 50
	Jump park
	Robot 2
	Motor On
	Power High
	Speed 50
	Accel 50, 50
	Jump park
	MemOff PartsToPick1
	MemOff PartsToPick4
	MemOff PartsToPick5

	PF_Start 1,4,5
	Xqt Robot1PickPlace
	Xqt Robot2PickPlace
Fend

Function Robot1PickPlace
	Robot 1
	Do
		Wait MemSw(PartsToPick1) = On
		PF_AccessFeeder(1)
		P0 = PF_QueGet(1)
		PF_QueRemove (1)
		Jump P0 /R
		On rbt1Gripper; Wait .25
		Jump place ! D30; MemOff PartsToPick1; PF_ReleaseFeeder 1!
		Off rbt1Gripper
		Wait .25
	Loop
Fend

Function Robot2PickPlace
	Robot 2
	Do
		Wait MemSw(PartsToPick4) = On
		PF_AccessFeeder(1)
		P0 = PF_QueGet(4)
		PF_QueRemove (4)
		Jump P0 /L
		On rbt2Gripper; Wait .25
		Jump place ! D30; MemOff PartsToPick4!
		Off rbt2Gripper; Wait .25
		Wait MemSw(PartsToPick5) = On
		P0 = PF_QueGet(5)
		PF_QueRemove (5)
		Jump P0 /L
		On rbt2Gripper; Wait .25
		Jump place ! D30; MemOff PartsToPick5; PF_ReleaseFeeder 1!
		Off rbt2Gripper; Wait .25
	Loop
Fend

PartFeeding.prg
Function PF_Robot(PartID As Integer) As Integer
	Select PartID
		Case 1
			MemOn PartsToPick1
			Wait MemSw(PartsToPick1) = Off
			PF_ActivePart 4
		Case 4
			MemOn PartsToPick4
			Wait MemSw(PartsToPick4) = Off
			PF_ActivePart 5
		Case 5
			MemOn PartsToPick5
			Wait MemSw(PartsToPick5) = Off
			PF_ActivePart 1
	Send
	PF_Robot = PF_CALLBACK_SUCCESS
Fend 

